***********************************************************************************************
																							  *
* Leadership Decapitation and Militant Alliance Breakdown							 		  *
																							  *
* Authors: Christopher W. Blair, Michael C. Horowitz, and Philip B.K. Potter				  *
* 																						      *
* September 2020																		      *
																							  *
***********************************************************************************************

clear all
set more off
set scheme plotplainblind
macro drop _all
estimates drop _all
set matsize 800
set seed 8675309

cd "/Users/christopherblair/Dropbox/Publications/Journal of Politics 2020/Data, Figures, and Models"

use "alliance_decapitation_final.dta"
estimates drop _all

sort dyad_id year
xtset dyad_id year

global org "sharedideo2 sharedsponsor_lag1 age1 age2 agediff capratioplus_dir grp1_newalliance_lag1 grp2_newalliance_lag1"

global context "ln_dist ln_pop1_lag1 ln_pop2_lag1 ln_gdppc1_lag1 ln_gdppc2_lag1 polity1_lag1 polity2_lag1 coldwar post911"

global dv "term1 collapse1 breakdown"

global iv "leaderkilled_grp1_lag1 leadercaptured_grp1_lag1 mutual_grp1_lag1 natcauses_grp1_lag1 otherexit_grp1_lag1 decapitation_grp1_lag1 thrownout_grp1_lag1 founderkilled_grp1_lag1 foundercaptured_grp1_lag1 founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1"

global cr "age1 age2 agediff capratioplus_dir grp1_newalliance_lag1 grp2_newalliance_lag1"


****************************************** FIGURE 1: TRENDS OVER TIME **********************************************


preserve

gen term=1 if term1==1
gen bdown=1 if breakdown==1
gen clpse=1 if collapse1==1

gen decap=1 if decapitation_grp1 ==1
gen kill=1 if leaderkilled_grp1 ==1
gen capt=1 if capturecaptkill_grp1 ==1
gen other=1 if otherexit_grp1 ==1

graph bar (count) bdown clpse if year<=2008, over(year) stack ylabel(0(5)30, angle(h) tlength(*1.45)) ymtick(##5, tlength(*1.75)) ytitle("# of Alliance Terminations") asyvars bar(1, fcolor(cranberry)) bar(2, fcolor("102 153 255")) legend(ring(0) position(1) rows(2) label(1 "Alliance Splits") label(2 "Group Collapse")) saving(terminationovertime.gph, replace)

graph bar (count) kill capt other if year<=2008, over(year) stack ylabel(0(5)30, angle(h) tlength(*1.45)) ymtick(##5, tlength(*1.75)) ytitle("# of Alliances Subject to Leadership Removal") asyvars bar(1, fcolor(black)) bar(2, fcolor(gs7)) bar(3, fcolor(gs13)) legend(ring(0) position(10) rows(2) label(1 "Leader Killed") label(2 "Leader Captured") label(3 "Other Exit")) saving(removalovertime.gph, replace)

graph combine "terminationovertime.gph" "removalovertime.gph", iscale (1.75*) saving(overtime.gph, replace)

restore


****************************************** FIGURE 2 and S.4: KM PLOTS **********************************************


stset term1_end, failure(term1) 

sts graph, ci failure by(decapitation_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Leader Decapitated") label(6 "Leader Decapitated")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Leader Decapitated") saving(decapitation.gph, replace)

sts graph, ci failure by(founderdecapitation_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Founder Decapitated") label(6 "Founder Decapitated")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Founder Decapitated") saving(fdecapitation.gph, replace)

sts graph, ci failure by(leaderkilled_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Leader Killed") label(6 "Leader Killed")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Leader Killed") saving(killed.gph, replace)

sts graph, ci failure by(founderkilled_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Founder Killed") label(6 "Founder Killed")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Founder Killed") saving(fkilled.gph, replace)

sts graph, ci failure by(leadercaptured_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Leader Captured") label(6 "Leader Captured")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Leader Captured") saving(captured.gph, replace)

sts graph, ci failure by(foundercaptured_grp1_lag1) plot1(lcolor(cranberry)) plot2(lcolor("102 153 255")) ci1opts(color(cranberry%20)) ci2opts(color("102 153 255"%30)) legend(ring(0) position(1) rows(4) label(1 "95% CI") label(3 "95% CI") label(5 "No Founder Captured") label(6 "Founder Captured")) xtitle("Years Since Alliance Start") ytitle("Probability of Alliance Termination") ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##2, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Founder Captured") saving(fcaptured.gph, replace)


************************************** TABLE 1: CONTINGENCY TABLE **********************************************


count if term1==1
count if term1==1 & decapitation_grp1_lag1==1
count if term1==1 & decapitation_grp1_any3==1
tabulate term1 decapitation_grp1_lag1, chi2
tabulate term1 decapitation_grp1_any3, chi2 

count if breakdown==1
count if breakdown==1 & decapitation_grp1_lag1==1
count if breakdown==1 & decapitation_grp1_any3==1
tabulate breakdown decapitation_grp1_lag1, chi2
tabulate breakdown decapitation_grp1_any3, chi2 

count if collapse1==1
count if collapse1==1 & decapitation_grp1_lag1==1
count if collapse1==1 & decapitation_grp1_any3==1
tabulate collapse1 decapitation_grp1_lag1, chi2
tabulate collapse1 decapitation_grp1_any3, chi2 


************************************** TABLE S.2: SUMMARY STATISTICS **********************************************


estpost sum $dv $iv $org $context

esttab using "sumstat.tex", label title(Summary Statistics) cells("count mean sd min max")


*************************************** CONDITIONAL FRAILTY *********************************************


/* 

Must be executed in R. See Final Alliance Decapitation Models.R for the R code.

*/



*************************************** TABLE S.7: COMPETING RISKS ************************************************


/* 

The variable competetype defines the competing termination risks, with:

	competetype==0 indicating no termination
	competetype==1 indicating termination by interorganizational split
	competetype==2 indicating termination by targeted group collapse
	
*/


** Unweighted Competing Risks Models -- Targeted Group Collapse as Outcome

stset collapse1_end, failure(competetype==2)

eststo cr1: stcrreg decapitation_grp1_lag1 $cr $context if year>=1971, compete(competetype==1) noshr vce(cluster dyad_id)

eststo cr2: stcrreg founderdecapitation_grp1_lag1 $cr $context if year>=1971, compete(competetype==1) noshr vce(cluster dyad_id)


** Unweighted Competing Risks Models -- Interorganization Split as Outcome

stset breakdown_end, failure(competetype==1)

eststo cr3: stcrreg decapitation_grp1_lag1 $org $context if year>=1971, compete(competetype==2) noshr vce(cluster dyad_id)

eststo cr4: stcrreg founderdecapitation_grp1_lag1 $org $context if year>=1971, compete(competetype==2) noshr vce(cluster dyad_id)

eststo cr5: stcrreg nonfounderdecapitation_lag1 $org $context if year>=1971, compete(competetype==2) noshr vce(cluster dyad_id)

eststo cr6: stcrreg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1 $org $context if year>=1971, compete(competetype==2) noshr vce(cluster dyad_id)

esttab cr1 cr2 cr3 cr4 cr5 cr6 using "/Users/christopherblair/Desktop/crisks.tex", t(3) legend label collabels(none) varlabels(_cons Constant) se(3) pr2 b(3) star(* 0.05 ** 0.01 *** 0.001) mtitles("C" "C" "S" "S" "S" "S") replace


****************************************** FIGURE S.6: CUMULATIVE INCIDENCE FUNCTIONS **********************************************


stset collapse1_end, failure(competetype==2)
est resto cr2
eststo cif2: stcrreg founderdecapitation_grp1_lag1 $cr $context, compete(competetype==1) noshr vce(cluster dyad_id)

stcurve, cif at1(founderdecapitation_grp1_lag1=1) at1(founderdecapitation_grp1_lag1=0) legend(ring(0) position(10) rows(2) label(1 "Founder Decapitated") label(2 "No Founder Decapitated") textw(small)) lcolor(cranberry "102 153 255") lpat(solid dash) ylabel(0 "0" 0.005 "0.005" 0.01 "0.01" 0.015 "0.015" 0.02 "0.02", angle(h) tlength(*1.45)) ymtick(##1, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Failure by Group Collapse") xtitle(Years Since Alliance Start) saving(cif1.gph, replace) 

stset breakdown_end, failure(competetype==1)

est resto cr4
stcurve, cif at1(founderdecapitation_grp1_lag1=1) at1(founderdecapitation_grp1_lag1=0) legend(ring(0) position(10) rows(2) label(1 "Founder Decapitated") label(2 "No Founder Decapitated") textw(small)) lcolor(cranberry "102 153 255") lpat(solid dash) ylabel(0 "0" 0.1 "0.1" .2 "0.2" .3 "0.3" .4 "0.4" .5 "0.5" .6 "0.6" .7 "0.7" .8 "0.8" .9 "0.9" 1 "1", angle(h) tlength(*1.45)) ymtick(##1, tlength(*1.75)) xlabel(0(2)40, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.75)) title("Failure by Interorganizational Split") xtitle(Years Since Alliance Start) saving(cif2.gph, replace)

graph combine "cif1.gph" "cif2.gph", cols(1) iscale(1.75*) saving(cif.gph, replace)



*********************************** TABLE S.8 and FIGURE S.9: CONSTRUCTING IPT WEIGHTS ***************************************


/* 

global denom "national1 left1 right1 environ1 sunnijihad1 multicountry1 ethfrac1 ln_attack1_lag1 age1 ln_pop1_lag1 ln_gdppc1_lag1 grp1_rev_fariss_lag1 majpower1_lag1 polity1_lag1 cwinc1_lag1"

global num "national1 left1 right1 environ1 sunnijihad1 multicountry1 ethfrac1"


logit decapitation_grp1 $denom, cluster(group1_id)
outreg2 using "/Users/christopherblair/Desktop/weights.tex", tex(land) label dec(3) replace ctitle(Denominator) keep($denom) alpha(0.01, 0.05, 0.1) addstat("Log Pseudo-Likelihood", e(ll))

predict denom if e(sample)
replace denom=denom*decapitation_grp1+(1-denom)*(1-decapitation_grp1)
sort dyad_id year
by dyad_id: replace denom=denom*denom[_n-1] if _n!=1
sum denom, detail


logit decapitation_grp1 $num, cluster(group1_id)
outreg2 using "/Users/christopherblair/Desktop/weights.tex", tex(land) label dec(3) append ctitle(Numerator) keep($num) alpha(0.01, 0.05, 0.1) addstat("Log Pseudo-Likelihood", e(ll))

predict num if e(sample)
replace num=num*decapitation_grp1+(1-num)*(1-decapitation_grp1)
sort dyad_id year
by dyad_id: replace num=num*num[_n-1] if _n!=1
sum num, detail

gen weights=num/denom
sum weights, detail 

graph box weights, over(year) noout saving(weights.gph, replace)

*/



************************************** FIGURE S.16: PARAMETRIC MODELS ******************************************


** Unweighted Exponential Distribution

stset term1_end, failure(term1)

eststo uwexp1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwexp2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwexp3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwexp4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

** Weighted Exponential Distribution

stset term1_end [pw=weights], failure(term1)

eststo wexp1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo wexp2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo wexp3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

eststo wexp4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(exponential) frailty(gamma) nohr vce(cluster dyad_id)

esttab uwexp1 uwexp2 uwexp3 uwexp4 wexp1 wexp2 wexp3 wexp4 using "/Users/christopherblair/Desktop/exp.tex", t(3) legend label collabels(none) varlabels(_cons Constant) se(3) pr2 b(3) star(* 0.05 ** 0.01 *** 0.001) replace


** Unweighted Weibull Distribution

stset term1_end, failure(term1)

eststo uwweib1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwweib2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwweib3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwweib4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

** Weighted Weibull Distribution

stset term1_end [pw=weights], failure(term1)

eststo wweib1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo wweib2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo wweib3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

eststo wweib4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(weibull) frailty(gamma) nohr vce(cluster dyad_id)

esttab uwweib1 uwweib2 uwweib3 uwweib4 wweib1 wweib2 wweib3 wweib4 using "/Users/christopherblair/Desktop/weib.tex", t(3) legend label collabels(none) varlabels(_cons Constant) se(3) pr2 b(3) star(* 0.05 ** 0.01 *** 0.001) replace


** Unweighted Gompertz Distribution

stset term1_end, failure(term1)

eststo uwgomp1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwgomp2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwgomp3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo uwgomp4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

** Weighted Gompertz Distribution

stset term1_end [pw=weights], failure(term1)

eststo wgomp1: streg decapitation_grp1_lag1 $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo wgomp2: streg founderdecapitation_grp1_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo wgomp3: streg nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

eststo wgomp4: streg founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1  $org $context if year>=1971, distribution(gompertz) frailty(gamma) nohr vce(cluster dyad_id)

esttab uwgomp1 uwgomp2 uwgomp3 uwgomp4 wgomp1 wgomp2 wgomp3 wgomp4 using "/Users/christopherblair/Desktop/gomp.tex", t(3) legend label collabels(none) varlabels(_cons Constant) se(3) pr2 b(3) star(* 0.05 ** 0.01 *** 0.001) replace


** Unweighted Logistic

eststo uwlog1: melogit term1 decapitation_grp1_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 || dyad_id:, vce(cluster dyad_id)

eststo uwlog2: melogit term1 founderdecapitation_grp1_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 || dyad_id:, vce(cluster dyad_id)

eststo uwlog3: melogit term1 nonfounderdecapitation_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 || dyad_id:, vce(cluster dyad_id)

eststo uwlog4: melogit term1 founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 || dyad_id:, vce(cluster dyad_id)

** Weighted Logistic

eststo wlog1: melogit term1 decapitation_grp1_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 [pw=weights] || dyad_id:, vce(cluster dyad_id)

eststo wlog2: melogit term1 founderdecapitation_grp1_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 [pw=weights] || dyad_id:, vce(cluster dyad_id)

eststo wlog3: melogit term1 nonfounderdecapitation_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 [pw=weights] || dyad_id:, vce(cluster dyad_id)

eststo wlog4: melogit term1 founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1 $org $context term1_end c.term1_end#c.term1_end c.term1_end#c.term1_end#c.term1_end if year>=1971 [pw=weights] || dyad_id:, vce(cluster dyad_id)

esttab uwlog1 uwlog2 uwlog3 uwlog4 wlog1 wlog2 wlog3 wlog4 using "/Users/christopherblair/Desktop/log.tex", t(3) legend label collabels(none) varlabels(_cons Constant) se(3) pr2 b(3) star(* 0.05 ** 0.01 *** 0.001) replace


**** Robustness Figure

coefplot (uwexp4, connect() lpatt(solid)lcol(black)msymbol(O)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) (wexp4, connect() lpatt(solid)lcol(black)msymbol(S)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black)))  || (uwweib4, connect() lpatt(solid)lcol(black)msymbol(O)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) (wweib4, connect() lpatt(solid)lcol(black)msymbol(S)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) || (uwgomp4, connect() lpatt(solid)lcol(black)msymbol(O)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) (wgomp4, connect() lpatt(solid)lcol(black)msymbol(S)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) || (uwlog4, connect() lpatt(solid)lcol(black)msymbol(O)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) (wlog4, connect() lpatt(solid)lcol(black)msymbol(S)msize(medlarge)mcol(black)ciopts(lpatt(solid)lcol(black))) , keep(founderdecapitation_grp1_lag1 nonfounderdecapitation_lag1) ci(95 90) yscale(off) xtitle("Standardized Coefficient") xline(0, lcolor(cranberry) lpattern(- -)) legend(off) xlabel(-3(1)2, angle(h) tlength(*1.45)) xmtick(##2, tlength(*1.55))

